

# NAG Fortran Library Routine Document

## F08FFF (SORGTR/DORGTR)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F08FFF (SORGTR/DORGTR) generates the real orthogonal matrix  $Q$ , which was determined by F08FEF (SSYTRD/DSYTRD) when reducing a symmetric matrix to tridiagonal form.

### 2 Specification

```

SUBROUTINE F08FFF(UPLO, N, A, LDA, TAU, WORK, LWORK, INFO)
ENTRY      sorgtr (UPLO, N, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER   N, LDA, LWORK, INFO
real    A(LDA,*), TAU(*), WORK(*)
CHARACTER*1 UPLO

```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine is intended to be used after a call to F08FEF (SSYTRD/DSYTRD), which reduces a real symmetric matrix  $A$  to symmetric tridiagonal form  $T$  by an orthogonal similarity transformation:  $A = QTQ^T$ . F08FEF (SSYTRD/DSYTRD) represents the orthogonal matrix  $Q$  as a product of  $n - 1$  elementary reflectors.

This routine may be used to generate  $Q$  explicitly as a square matrix.

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: UPLO – CHARACTER\*1 *Input*  
*On entry:* this **must** be the same parameter UPLO as supplied to F08FEF (SSYTRD/DSYTRD).  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $Q$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,\*) – **real** array *Input/Output*  
**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .  
*On entry:* details of the vectors which define the elementary reflectors, as returned by F08FEF (SSYTRD/DSYTRD).  
*On exit:* the  $n$  by  $n$  orthogonal matrix  $Q$ .

- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array A as declared in the (sub)program from which F08FFF (SORGTR/DORGTR) is called.  
*Constraint:*  $LDA \geq \max(1, N)$ .
- 5: TAU(\*) – *real* array *Input*  
**Note:** the dimension of the array TAU must be at least  $\max(1, N - 1)$ .  
*On entry:* further details of the elementary reflectors, as returned by F08FEF (SSYTRD/DSYTRD).
- 6: WORK(\*) – *real* array *Workspace*  
**Note:** the dimension of the array WORK must be at least  $\max(1, LWORK)$ .  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimum performance.
- 7: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08FFF (SORGTR/DORGTR) is called, unless LWORK = -1, in which case a workspace query is assumed and the routine only calculates the optimal dimension of WORK (using the formula given below).  
*Suggested value:* for optimum performance LWORK should be at least  $(N - 1) \times nb$ , where *nb* is the *blocksize*.  
*Constraint:*  $LWORK \geq \max(1, N - 1)$  or LWORK = -1.
- 8: INFO – INTEGER *Output*  
*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If  $INFO = -i$ , the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

## 7 Accuracy

The computed matrix *Q* differs from an exactly orthogonal matrix by a matrix *E* such that

$$\|E\|_2 = O(\epsilon),$$

where  $\epsilon$  is the *machine precision*.

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{4}{3}n^3$ .

The complex analogue of this routine is F08FTF (CUNGTR/ZUNGTR).

## 9 Example

To compute all the eigenvalues and eigenvectors of the matrix  $A$ , where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix}.$$

Here  $A$  is symmetric and must first be reduced to tridiagonal form by F08FEF (SSYTRD/DSYTRD). The program then calls F08FFF (SORGTR/DORGTR) to form  $Q$ , and passes this matrix to F08JEF (SSTEQR/DSTEQR) which computes the eigenvalues and eigenvectors of  $A$ .

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F08FFF Example Program Text
*      Mark 16 Release. NAG Copyright 1992.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, LDA, LWORK, LDZ
PARAMETER       (NMAX=8,LDA=NMAX,LWORK=64*NMAX,LDZ=NMAX)
*      .. Local Scalars ..
INTEGER          I, IFAIL, INFO, J, N
CHARACTER        UPLO
*      .. Local Arrays ..
real            A(LDA,NMAX), D(NMAX), E(NMAX), TAU(NMAX),
+               WORK(LWORK), Z(LDZ,NMAX)
*      .. External Subroutines ..
EXTERNAL         sorgtr, ssteqr, ssytrd, F06QFF, X04CAF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F08FFF Example Program Results'
Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
*      Read A from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF

*
*      Reduce A to tridiagonal form T = (Q**T)*A*Q
*
      CALL ssytrd(UPLO,N,A,LDA,D,E,TAU,WORK,LWORK,INFO)

*
*      Copy A into Z
*
      CALL F06QFF(UPLO,N,N,A,LDA,Z,LDZ)

*
*      Form Q explicitly, storing the result in Z
*
      CALL sorgtr(UPLO,N,Z,LDZ,TAU,WORK,LWORK,INFO)

*
*      Calculate all the eigenvalues and eigenvectors of A
*
      CALL ssteqr('V',N,D,E,Z,LDZ,WORK,INFO)

*
      WRITE (NOUT,*)
      IF (INFO.GT.0) THEN
        WRITE (NOUT,*) 'Failure to converge.'
```

```

      ELSE
*
*       Print eigenvalues and eigenvectors
*
      WRITE (NOUT,*) 'Eigenvalues'
      WRITE (NOUT,99999) (D(I),I=1,N)
      WRITE (NOUT,*)
      IFAIL = 0
*
      CALL X04CAF('General',' ',N,N,Z,LDZ,'Eigenvectors',IFAIL)
*
      END IF
      END IF
      STOP
*
99999 FORMAT (3X,(8F8.4))
      END

```

## 9.2 Program Data

```

F08FFF Example Program Data
  4                               :Value of N
  'L'                             :Value of UPLO
  2.07
  3.87 -0.21
  4.20  1.87  1.15
 -1.15  0.63  2.06 -1.81   :End of matrix A

```

## 9.3 Program Results

F08FFF Example Program Results

```

Eigenvalues
  -5.0034 -1.9987  0.2013  8.0008

```

```

Eigenvectors
      1      2      3      4
1  0.5658 -0.2328 -0.3965  0.6845
2 -0.3478  0.7994 -0.1780  0.4564
3 -0.4740 -0.4087  0.5381  0.5645
4  0.5781  0.3737  0.7221  0.0676

```

---